

## Funktionen - erweitert

### Excel 2010 - Fortgeschritten

Dagmar Serb

V.01/Jän. 2016

<b>FUNKTIONEN – ERWEITERT .....</b>	<b>1</b>
MATHEMATISCHE FUNKTIONEN .....	1
<i>DBSUMME</i> .....	1
STATISTISCHE FUNKTIONEN .....	2
<i>ZÄHLENWENN</i> .....	2
<i>ZÄHLENWENNNS</i> .....	3
TEXT-FUNKTIONEN .....	4
<i>LINKS</i> .....	4
<i>RECHTS</i> .....	4
<i>TEIL</i> .....	5
<i>ERSETZEN</i> .....	5
<i>WECHSELN</i> .....	6
<i>SUCHEN</i> .....	6
<i>FINDEN</i> .....	7
<i>VERKETTEN</i> .....	8
<i>TEXT</i> .....	9
<i>LÄNGE</i> .....	10
<i>BEISPIELE FÜR VERSCHACHTELUNGEN</i> .....	10
Mit <i>LINKS</i> und <i>FINDEN</i> Zeichen extrahieren .....	10
Mit <i>VERKETTEN</i> und <i>TEXT</i> Ergebnis in Satz ausformulieren .....	11
DATUMS-FUNKTIONEN .....	11
<i>NETTOARBEITSTAGE</i> .....	12
<i>ARBEITSTAG</i> .....	12
<i>ENDDATUM</i> .....	13

## Funktionen – erweitert

Die angeführten Beispiele finden Sie in der Übungsdatei Funktionen-Erweitert.xlsx

### MATHEMATISCHE Funktionen

#### DBSUMME

Mit dieser Funktion kann man **Summen bilden**, die **bestimmten Suchkriterien** entsprechen.

Syntax	
DBSUMME(Datenbank;Datenbankfeld;Suchkriterien)	

Argumente	Beschreibung
<b>Datenbank</b> (erforderlich)	Der <b>Zellbereich, aus dem die Liste oder Datenbank besteht</b> . Die erste Zeile der Liste enthält <b>Spaltenbeschriftungen</b> .
<b>Datenbankfeld</b> (erforderlich)	Gibt an, <b>welche Spalte in der Funktion verwendet wird</b> . Sie können dafür die <b>Spaltenbeschriftung in Anführungszeichen</b> angeben (z.B. „Rabatt“ <b>oder eine Zahl</b> (ohne Anführungszeichen), <b>die die Position der Spalte in der Liste darstellt</b> : 1 für die erste Spalte, 2 für die zweite Spalte usw.
<b>Suchkriterien</b> (erforderlich)	Der <b>Zellbereich, der die angegebenen Bedingungen enthält</b> . Sie können jeden Bereich verwenden, der mindestens eine Spaltenbeschriftung und eine Zelle unter der Beschriftung zum Angeben der Bedingung enthält.

Beispiel: Alle Umsätze der Kunden aus Deutschland, mit denen ein Jahresumsatz von mehr als 500.000 € erzielt wurde und dabei die Anzahl der Aufträge bei mehr als 5 liegt, sollen in Zelle C10 summiert werden:

	A	B	C	D
1	Kunde	Land	Jahresumsatz	Aufträge
2	Kaiser	Deutschland	€ 750.567,00	6
3	Beckmann	Schweiz	€ 289.675,00	3
4	Müller	Deutschland	€ 118.514,00	2
5	Rickmers	Deutschland	€ 861.536,00	5
6				
7	Analyse		Datenbankfeld	
8	Kunde	Land	Jahresumsatz	Aufträge
9		Deutschland	>500000	>5
10			€ 750.567,00	

- Für das erste **Argument „Datenbank“** wird der **Zellbereich A1 bis D5** (Datenbank inkl. Spaltenüberschriften!) eingegeben.
- Das zweite **Argument „Datenbankfeld“** ist **Zelle C8** (Jahresumsatz), welches summiert werden soll. Kann in der Formel auch als Text eingetragen werden: "Jahresumsatz".
- **„Suchkriterien“** ist der **Zellbereich A8 bis D9** (Darauf achten, dass die Überschriften sowohl im Daten- als auch im Kriterienbereich von den Feldnamen her gleich sind. Es müssen nicht alle Felder aus dem Datenbereich auch im Kriterienbereich angegeben werden!)

## STATISTISCHE Funktionen

### ZÄHLENWENN

Diese Funktion **zählt alle Werte**, die das **Suchkriterium erfüllen**.

Syntax
<b>=ZÄHLENWENN(Bereich; Suchkriterium)</b>

Argumente	Beschreibung
<b>Bereich</b> (erforderlich)	<b>Bereich</b> , in dem die <b>Zellen gezählt</b> werden. Leere Zellen und Textwerte werden ignoriert.
<b>Suchkriterium</b> (erforderlich)	Suchkriterium, <b>welches im Bereich gezählt wird</b> . Es kann eine Zahl, ein Ausdruck, ein Zellbezug oder eine Textzeichenfolge sein. Z.B.: <i>Zahl 32, Vergleich "&gt;32", Bezug B2 oder "Text"</i> .

Ein paar Beispiele:

	A	B	C	D	E	F	G
1	Sorte	Kisten					
2	Apfel	12					
3	Birne	16					
4	Kirsche	8					
5	Apfel	36					
6	Birne	13					
7	Apfel	14					
8							
a) 9	Anzahl der Zellen mit dem Wert Apfel:			3	=ZÄHLENWENN(A2:A7;"Apfel")		
b) 10	Anzahl der Zellen mit einem Wert größer als 15:			2	=ZÄHLENWENN(B2:B7;">15")		
c) 11	Anzahl der Zellen mit dem Wert Apfel und Birne:			5	=ZÄHLENWENN(A2:A7;A2)+ZÄHLENWENN(A2:A7;A3)		
d) 12	Anzahl der Zellen mit beliebigem Text:			6	=ZÄHLENWENN(A2:A7;"*")		
e) 13	Anzahl der Zellen mit 5 Zeichen und "el" als letzten beiden Zeichen:			3	=ZÄHLENWENN(A2:A7;"???el")		

- „Bereich“**: Gezählt wird im Bereich **A2 bis A7 (Sorten)**, und zwar die Sorte „Apfel“, **„Suchkriterium“** ist daher **"Apfel"**. Das Ergebnis lautet „3“.
- „Bereich“**: Gezählt wird im Bereich **B2 bis B7 (Anzahl Kisten)**, und zwar alle Werte größer als 15, **„Suchkriterium“** ist daher **">15"**. Das Ergebnis lautet „2“.
- „Bereich“**: Gezählt wird im Bereich **A2 bis A7 (Sorten)**, und zwar die Sorte „Apfel“, das **„Suchkriterium“** wird diesmal als **Zellbezug A2** („Apfel“) angegeben. Weiters muss nach gleichem Schema nach „Birne“ gesucht werden. Diese **zweite Funktion wird mit einem Pluszeichen (+) an die erste Funktion angehängt**. Das Ergebnis lautet „5“.
- „Bereich“**: Gezählt wird im Bereich **A2 bis A7 (Sorten)**, und zwar nach beliebigem Text, das **„Suchkriterium“** lautet daher **"\*"** („\*“ wird als Platzhalterzeichen für eine beliebige Zeichenfolge verwendet). Das Ergebnis lautet „6“.
- „Bereich“**: Gezählt wird im Bereich **A2 bis A7 (Sorten)**, und zwar nach Zellen mit 5 Zeichen und „el“ als letzte beiden Zeichen, **„Suchkriterium“** ist daher **"???el"**. („?“ wird als Platzhalterzeichen für ein einzelnes Zeichen verwendet). Das Ergebnis lautet „3“

## ZÄHLENWENNS

Diese Funktion **zählt alle Werte**, die die **Suchkriterien erfüllen**.

Syntax
<b>=ZÄHLENWENNS( Kriterienbereich1;Kriterien1;[Kriterienbereich2; Kriterien2]...)</b>

Argumente	Beschreibung
<b>Kriterienbereich 1</b> (erforderlich)	<b>Erster Bereich</b> , in dem die <b>zugehörigen Kriterien ausgewertet</b> werden.
<b>Kriterien 1</b> (erforderlich)	Kriterium, <b>welches im entsprechenden Bereich gezählt wird</b> . Es kann eine Zahl, ein Ausdruck, ein Zellbezug oder eine Textzeichenfolge sein. Z.B.: <i>Zahl 32, Vergleich "&gt;32", Bezug B2 oder "Text"</i> .
<b>Kriterienbereich2;Kriterien2;</b> ... (optional)	Zusätzliche Bereiche und deren zugehörige Kriterien. Es sind bis zu 127 Bereich/Kriterien-Paare zulässig.

In diesem Beispiel wird gezählt, wie viele Verkäufer sowohl Bonus 1 als auch Bonus 2 erhalten:

D7	fx =ZÄHLENWENNS(B2:B5;"ja";C2:C5;"ja")			
	A	B	C	D
1	Verkäufer	Bonus 1	Bonus 2	Bonus 3
2	Maier	ja	nein	nein
3	Müller	ja	ja	nein
4	Hofer	ja	ja	ja
5	Bauer	nein	ja	ja
6				
7	Wie viele Verkäufer erhalten sowohl Bonus 1 als auch Bonus 2:			2

- „**Kriterienbereich 1**“ ist **B2 bis B5** (Werte von Bonus1), „**Kriterien 1**“ ist **"ja"**.
- „**Kriterienbereich 2**“ ist **C2 bis C5** (Werte von Bonus2), „**Kriterien 2**“ ist **"ja"**. Das Ergebnis lautet „2“.

Dieses Beispiel ermittelt, wie viele Regionen im Jänner mehr als 25.000 Euro und im März mehr als 23.000 Euro generieren konnten:

D6	fx =ZÄHLENWENNS(B2:B4;">25000";D2:D4;">23000")				
	A	B	C	D	E
1	Verkäufer	Jänner	Februar	März	
2	Region 1	€ 26.000,00	€ 25.000,00	€ 27.000,00	
3	Region 2	€ 29.000,00	€ 25.000,00	€ 21.000,00	
4	Region 3	€ 23.000,00	€ 30.000,00	€ 24.000,00	
5					
6	Wie viele Regionen konnten im Jänner mehr als 25.000 Euro und im März mehr als 23.000 Euro generieren:			1	

- „**Kriterienbereich 1**“ ist **B2 bis B4** (Werte Jänner), „**Kriterien 1**“ ist **">25000"**.
- „**Kriterienbereich 2**“ ist **D2 bis D4** (Werte März), „**Kriterien 2**“ ist **">23000"**. Das Ergebnis lautet „1“.

## TEXT-Funktionen

Textfunktionen können **separat** verwendet werden, kommen in der Praxis aber häufig **in Kombination miteinander** vor. Sogenannte Verschachtelungen braucht man zum Beispiel dazu, um aus Listen **Daten herauszuziehen** (*PLZ und Ort trennen*). Zunächst aber die Funktionen einzeln erklärt:

### LINKS

Diese Funktion **gibt** je nach Angabe **das oder die erste(n) Zeichen** einer Textzeichenfolge zurück.

Syntax
=LINKS(Text;[Anzahl_Zeichen])

Argumente	Beschreibung
<b>Text</b> (erforderlich)	<b>Text</b> , der zu extrahieren ist.
<b>Anzahl_Zeichen</b> (optional)	Gibt die <b>Anzahl</b> der <b>Zeichen</b> an, die zu <b>extrahieren</b> sind. <ul style="list-style-type: none"> <li>Anzahl_Zeichen muss <b>größer gleich 0</b> sein.</li> <li>Ist Anzahl_Zeichen größer als „Text“, wird der gesamte Text zurückgegeben.</li> <li>Fehlt Anzahl_Zeichen, wird es als 1 angenommen.</li> </ul>

Beispiel:

B2 <span style="border: 1px solid red; padding: 2px;">=LINKS(A2;4)</span>				
	A	B	C	D
1	<b>Daten</b>	<b>Ergebnis</b>	<b>Formel</b>	<b>Beschreibung</b>
2	8700 Leoben	8700	=LINKS(A2;4)	Extrahiert die ersten vier Zeichen der Zeichenfolge in A2.
3	8600 Bruck/Mur	8600	=LINKS(A3;4)	Extrahiert die ersten vier Zeichen der Zeichenfolge in A3.
4	8650 Kindberg	8650	=LINKS(A4;4)	Extrahiert die ersten vier Zeichen der Zeichenfolge in A4.

### RECHTS

Diese Funktion **gibt** je nach Angabe **das oder die letzte(n) Zeichen einer Textzeichenfolge zurück**.

Syntax
=RECHTS(Text;[Anzahl_Zeichen])

Argumente	Beschreibung
<b>Text</b> (erforderlich)	<b>Text</b> , der zu extrahieren ist.
<b>Anzahl_Zeichen</b> (optional)	Gibt die <b>Anzahl</b> der <b>Zeichen</b> an, die zu <b>extrahieren</b> sind. <ul style="list-style-type: none"> <li>Anzahl_Zeichen muss <b>größer gleich 0</b> sein.</li> <li>Ist Anzahl_Zeichen größer als „Text“, wird der gesamte Text zurückgegeben.</li> <li>Fehlt Anzahl_Zeichen, wird es als 1 angenommen.</li> </ul>

Beispiel:

C2		fx =RECHTS(A2;3)		
	A	B	C	D
1	Daten	Ergebnis	Formel	Beschreibung
2	Artikel001	001	=RECHTS(A2;3)	Extrahiert die letzten drei Zeichen der Zeichenfolge in A2 .
3	Artikel002	002	=RECHTS(A3;3)	Extrahiert die letzten drei Zeichen der Zeichenfolge in A3 .
4	Artikel003	003	=RECHTS(A4;3)	Extrahiert die letzten drei Zeichen der Zeichenfolge in A4 .

## TEIL

Diese Funktion **extrahiert** aus einem Text je nach Angabe **eines oder mehrere Zeichen**.

Syntax	
=TEIL(Text;Erstes_Zeichen;Anzahl_Zeichen)	

Argumente	Beschreibung
<b>Text</b> (erforderlich)	<b>Text</b> , der zu extrahieren ist.
<b>Erstes_Zeichen</b> (erforderlich)	Gibt an, <b>ab welchem Zeichen</b> der Inhalt extrahiert wird.
<b>Anzahl_Zeichen</b> (erforderlich)	Gibt die <b>Anzahl der zu extrahierenden Zeichen</b> an.

Beispiel:

B2		fx =TEIL(A2;6;50)		
	A	B	C	D
1	Daten	Ergebnis	Formel	Beschreibung
2	8700 Leoben	Leoben	=TEIL(A2;6;50)	Extrahiert aus der Zeichenfolge in A2 ab dem 6. Zeichen 50 Zeichen.
3	8600 Bruck/Mur	Bruck/Mur	=TEIL(A3;6;50)	Extrahiert aus der Zeichenfolge in A3 ab dem 6. Zeichen 50 Zeichen.
4	8650 Kindberg	Kindberg	=TEIL(A4;6;50)	Extrahiert aus der Zeichenfolge in A4 ab dem 6. Zeichen 50 Zeichen.

## ERSETZEN

Diese Funktion **ersetzt** innerhalb eines Texts eine **an einer bestimmten Position beginnende Zeichenfolge**. (Im Unterschied dazu tauscht WECHSELN innerhalb eines Texts eine bestimmte Zeichenfolge aus).

Syntax	
=ERSETZEN(Alter_Text;Erstes_Zeichen;Anzahl_Zeichen;Neuer-Text)	

Argumente	Beschreibung
<b>Alter_Text</b> (erforderlich)	<b>Text</b> , bei dem eine Anzahl von Zeichen ersetzt wird.
<b>Erstes_Zeichen</b> (erforderlich)	<b>Position des Zeichens</b> in "Alter_Text", an der mit dem Ersetzen begonnen wird.
<b>Anzahl_Zeichen</b> (erforderlich)	<b>Anzahl der Zeichen</b> in "Alter_Text", die durch "Neuer_Text" ersetzt werden.
<b>Neuer_Text</b> (erforderlich)	<b>Text</b> , der die Zeichen in "Alter_Text" <b>ersetzt</b> .

Beispiele:

B1 $f_x$ =ERSETZEN(A1;3;2;"15")			
	A	B	C
1	2014	2015	Ersetzt die beiden letzten Ziffern von 2014 durch 15
B1 $f_x$ =ERSETZEN(A1;9;1;"(at)")			
	A	B	C
1	m.muster@excel.at	m.muster(at)excel.at	Ersetzt das alte Zeichen von "m.muster@excel.at" durch "at".

## WECHSELN

Diese Funktion **ersetzt** innerhalb eines Textes **eine bestimmte Zeichenfolge** (Im Unterschied dazu ersetzt ERSETZEN innerhalb eines Textes eine an einer bestimmten Position beginnende Zeichenfolge).

Syntax	
WECHSELN(Text;Alter_Text;Neuer_Text;[ntes_Auftreten])	

Argumente	Beschreibung
<b>Text</b> (erforderlich)	<b>Text</b> (in Hochkommas (!)) <b>oder Zellbezug</b> , in dem Zeichen ausgetauscht werden.
<b>Alter_Text</b> (erforderlich)	<b>Text</b> , der <b>ersetzt</b> wird (in Hochkommas (!)).
<b>Neuer_Text</b> (erforderlich)	<b>Text</b> , der „Alter_Text“ <b>ersetzt</b> (in Hochkommas (!)).
<b>ntes_Auftreten</b> (optional)	Gibt an, <b>an welchen Stellen</b> „Alter_Text“ durch „Neuer_Text“ ersetzt wird. Wird „ntes_Auftreten“ <b>nicht angegeben</b> , werden <b>alle Vorkommen von „Alter_Text“ ersetzt</b> .

Beispiele:

B2 $f_x$ =WECHSELN(A2;"Umsatz";"Verkaufs")				
	A	B	C	D
1	<b>Daten</b>	<b>Ergebnis</b>	<b>Formel</b>	<b>Beschreibung</b>
2	Umsatzdaten	Verkaufsdaten	=WECHSELN(A2;"Umsatz";"Verkaufs")	Wechselt "Umsatz" in "Verkaufs".
3	Quartal 2, 2015	Quartal 3, 2015	=WECHSELN(A3;"2";"3";1)	Wechselt die erste Instanz von "2" in "3".
4	Quartal 1, 2011	Quartal 1, 2015	=WECHSELN(A4;"1";"5";3)	Wechselt die dritte Instanz von "1" in "5".

## SUCHEN

Diese Funktion **sucht nach dem ersten Vorkommen einer Zeichenfolge innerhalb einer anderen Zeichenfolge** und gibt dabei die **Position der Fundstelle als Nummer** zurück. SUCHEN **unterscheidet nicht zwischen Groß- und Kleinschreibung** und unterstützt die **Platzhalterzeichen** „?“ und „\*“ (Im Unterschied dazu unterscheidet FINDEN zwischen Groß- und Kleinschreibung und unterstützt keine Platzhalterzeichen.).

## Syntax

```
=SUCHEN(Suchtext;Text;Erstes_Zeichen)
```

Argumente	Beschreibung
<b>Suchtext</b> (erforderlich)	<b>Text</b> , der zu <b>finden</b> ist.
<b>Text</b> (erforderlich)	<b>Text</b> , in dem „Suchtext“ gesucht wird.
<b>Erstes_Zeichen</b> (optional)	Gibt an, <b>bei welchem Zeichen die Suche begonnen</b> wird. Das erste Zeichen hat die Nummer 1. Fehlt das Argument „Erstes_Zeichen“, wird es als 1 angenommen.

Beispiele:

B2 $f_x$ =SUCHEN("C";A2)		
	A	B
1	Text	Ergebnis Beschreibung
2	Frank Mc Court	8 Ermittelt die Position des ersten "C" in der Zeichenfolge. Ergebnis: "8"

B2 $f_x$ =SUCHEN("fahr";A2)		
	A	B
1	Text	Ergebnis Beschreibung
2	Fahrradfahren	1 Ermittelt die Position des ersten "fahr" in der Zeichenfolge. Ergebnis: "1"

B2 $f_x$ =SUCHEN("r?n";A2)		
	A	B
1	Text	Ergebnis Beschreibung
2	Fahrradfahren	11 Ermittelt die Position der ersten Zeichenfolge, die mit "r" beginnt und mit "n" endet. Dazwischen darf sich ein Zeichen befinden. Ergebnis: "11"

B2 $f_x$ =SUCHEN("r*n";A2)		
	A	B
1	Text	Ergebnis Beschreibung
2	Fahrradfahren	4 Ermittelt die Position der ersten Zeichenfolge, die mit "r" beginnt und mit "n" endet. Dazwischen dürfen sich mehrere Zeichen befinden. Ergebnis: "4"



## FINDEN

Diese Funktion **sucht nach dem ersten Vorkommen einer Zeichenfolge innerhalb einer anderen Zeichenfolge** und gibt dabei die **Position der Fundstelle als Nummer** zurück. **FINDEN unterscheidet zwischen Groß- und Kleinschreibung!** (Im Unterschied dazu unterscheidet SUCHEN nicht zwischen Groß- und Kleinschreibung und die Verwendung von Platzhalterzeichen ist möglich!)

**Syntax**  
**=FINDEN(Suchtext;Text;Erstes\_Zeichen)**

Argumente	Beschreibung
<b>Suchtext</b> (erforderlich)	<b>Text</b> , der zu <b>finden</b> ist.
<b>Text</b> (erforderlich)	<b>Text</b> , in dem „Suchtext“ gesucht wird.
<b>Erstes_Zeichen</b> (optional)	Gibt an, <b>bei welchem Zeichen die Suche begonnen</b> wird. Das erste Zeichen hat die Nummer 1. Fehlt das Argument „Erstes_Zeichen“, wird es als 1 angenommen.

Beispiele:


**VERKETTEN**

Diese Funktion **verbindet** zwei oder mehrere **Zeichenfolgen zu einer Zeichenfolge**.

**Syntax**  
**= VERKETTEN(Text1;[Text2];...)**

Argumente	Beschreibung
<b>Text1</b> (erforderlich)	Die erste <b>zu verknüpfende Zeichenfolge</b> .
<b>Text2, ...</b> (optional)	<b>Weitere</b> , bis zu 255 <b>verknüpfende Zeichenfolgen</b> .

Beispiele:



Verkettet wird **A2** (Frank) mit **B2** (Mc Court). Das **Leerzeichen** zwischen Vor- und Nachname wird in der Formel **zwischen Hochkommata (" ")** eingegeben.

D2	fx =VERKETTEN(A2;" ";B2;" ";ist ein fantastischer;" ";C2)			
	A	B	C	D
1	Daten	Daten	Daten	Ergebnis
2	Frank	Mc Court	Buchautor	Frank Mc Court ist ein fantastischer Buchautor

Verkettet wird **A2** (Frank) mit **B2** (Mc Court). Das **Leerzeichen** zwischen Vor- und Nachname wird in der Formel zwischen **Hochkommas** (" ") eingegeben. Es folgen ein **Leerzeichen** und der **Text** „ist ein fantastischer“ (in Hochkommas!). Nach einem weiteren **Leerzeichen** wird „Buchautor“ als **Zellbezug C2** eingefügt.

☺ **Tipps:** Mit dem **kaufmännischen Und-Zeichen (&)** als Berechnungsoperator können Sie **Textelemente verknüpfen**, ohne eine Funktion verwenden zu müssen!

C2	fx =A2&" "&B2				
	A	B	C	D	E
1	Daten	Daten	Ergebnis	Formel	Beschreibung
2	Frank	Mc Court	Frank Mc Court	=A2&" "&B2	A2 (Frank) und B2 (Mc Court) werden mit "&" verknüpft. Das Leerzeichen zwischen Vor- und Nachname wird in Hochkommas (" ") eingegeben.
3	H	M	H&M	=A3&" "&B3	A3 (H) und B3 (M) werden mit "&" verknüpft. Das "&" zwischen "H" und "M" wird in Hochkommas (" ") eingegeben.

## TEXT

Diese Funktion **wandelt einen numerischen Wert in Text um**, wobei gleichzeitig das Anzeigeformat festgelegt wird.

Syntax	
TEXT(Wert;Textformat)	

Argumente	Beschreibung
<b>Wert</b> (erforderlich)	Numerischer <b>Wert</b> , <b>Formel</b> , die einen numerischen Wert liefert, oder <b>Bezug</b> auf eine Zelle, die einen numerischen Wert enthält.
<b>Textformat</b> (erforderlich)	<b>Numerisches Format</b> als eine Zeichenfolge in Anführungszeichen (z. B. "t.M.jjjj" oder "#.#0,00").

Beispiele:

B2	fx =TEXT(A2;"#,000")			
	A	B	C	D
1	Daten	Ergebnis	Formel	Beschreibung
2	5,4	5,400	=TEXT(A2;"#,000")	Gibt den Wert in Zelle A2 mit drei Nachkommastellen aus.
3	29,895	29,90 €	=TEXT(A3;"#.###,00 €")	Gibt den Wert in Zelle A3 mit Tausendertrennpunkt, 2 Nachkommastellen und dem Eurozeichen aus.
4	5,25	5 1/4	=TEXT(A4;"# ??/? ??")	Gibt die Nachkommastelle des Werts in A4 als Bruch aus.
5	09.Dez	09.12.2015	=TEXT(A5;"tt.MM.jjjj")	Zeigt das Datum in A5 wie folgt an: Tag mit vorangestellter 0, Monat falls zutreffend mit vorangestellter 0, Jahreszahl vierstellig.

## LÄNGE

Diese Funktion gibt die **Anzahl der Zeichen** einer Zeichenfolge zurück.

Syntax
LÄNGE(Text)

Argumente	Beschreibung
<b>Text</b> (erforderlich)	<b>Text</b> , dessen Länge ermittelt werden soll. <i>Leerzeichen zählen als Zeichen.</i>

Beispiel:

B2		fx	
		=LÄNGE(A2)	
	A	B	C
1	Daten	Ergebnis	Formel
2	Meine Überschrift	17	=LÄNGE(A2)

## BEISPIELE FÜR VERSCHACHELUNGEN

### Mit LINKS und FINDEN Zeichen extrahieren

Im Übungsbeispiel der Funktion LINKS wurden Postleitzahlen von Orten getrennt. Ist die **Anzahl der zu extrahierenden Zeichen variabel** (österreichische PLZ sind 4stellig, deutsche 5stellig!), wird **zusätzlich** die **Funktion FINDEN** benötigt, um die **Stelle für die Extraktion zu ermitteln**.

B2		fx	
		=LINKS(A2;FINDEN("-",A2)-1)	
	A	B	C
1	Daten	Ergebnis	Formel
2	8700-Leoben	8700	=LINKS(A2;FINDEN("-",A2)-1)
3	20095-Hamburg	20095	=LINKS(A3;FINDEN("-",A3)-1)
4	8600-Bruck/Mur	8600	=LINKS(A4;FINDEN("-",A4)-1)

Die Funktion Schritt für Schritt:

- Die Funktion beginnt mit **=LINKS(**
- Das Argument „Text“ ist der **Zellbezug A2**:  
**=LINKS(A2;**
- Für das zweite Argument „Anzahl\_Zeichen“ müsste die Zahl der zu extrahierenden Zeichen eingegeben werden, diese ist in diesem Beispiel aber variabel (**4- und 5stellige PLZ**)! Daher kommt jetzt die zweite Funktion ins Spiel: **Die Funktion FINDEN ermittelt die Position des Bindestrichs zwischen PLZ und Ort, die als eindeutige Trennstelle gesehen werden kann**:  
**=LINKS(A2;FINDEN("-",A2)**
- Damit der **Bindestrich** nicht mit extrahiert wird, wird er anschließend **mit -1 abgezogen**:  
**=LINKS(A2;FINDEN("-",A2)-1**
- Nach dem Schließen der **Klammer** kann die Formel nach unten kopiert werden:  
**=LINKS(A2;FINDEN("-",A2)-1)**

### Mit VERKETTEN und TEXT Ergebnis in Satz ausformulieren

Im nächsten Beispiel wird der errechnete **Artikelpreis** mithilfe der Funktionen TEXT und VERKETTEN **als ausformulierter Satz wiedergegeben**.

	A	B	C	D	E	F	G
1	Preis Netto	MwSt.	Preis Brutto	Datum			
2	26	5 €	31 €	42347			
3							
4	Der Artikel kostet 31,20 € inkl. MwSt.						
5	Datum: 2015-12-09						

Die Funktion Schritt für Schritt:

- Die Funktion beginnt mit **=VERKETTEN(**
- Argument „Text 1“ ist der Text „Der Artikel kostet“  
**=VERKETTEN("Der Artikel kostet";**
- Argument „Text 2“ ist ein **Leerzeichen**  
**=VERKETTEN("Der Artikel kostet";" ");**
- Als Argument „Text 3“ wird der errechnete Bruttobetrag als **TEXT-Funktion** eingebunden und mit 2 Dezimalstellen und dem Eurozeichen formatiert:  
**=VERKETTEN("Der Artikel kostet";" ";TEXT(C2;"#.##0,00 €");**
- Argument „Text 4“ ist wieder ein **Leerzeichen**:  
**=VERKETTEN("Der Artikel kostet";" ";TEXT(C2;"#.##0,00 €");" ");**  
Argument „Text 5“ ist der Text **"inkl. MwSt."**  
**=VERKETTEN("Der Artikel kostet";" ";TEXT(C2;"#.##0,00 €");" ";inkl. MwSt."**
- Die Funktion wird mit **Klammer** abgeschlossen.  
**=VERKETTEN("Der Artikel kostet";" ";TEXT(C2;"#.##0,00 €");" ";inkl. MwSt.")**

👁 **Hinweis:** Sie können natürlich auch auf die Funktion VERKETTEN verzichten und **mit dem kaufmännischen Und (&) arbeiten:**

**=**"Der Artikel kostet "& TEXT(C2;"#.##0,00 €") & " inkl. MwSt."

Auf gleiche Weise kann das Datum (Zelle D2) ausformuliert werden:

**=**"Datum: " & TEXT(D2;"jjjj-MM-tt") ergibt: Datum: 2015-12-09

## DATUMS-Funktionen

In Excel wird die **Zahl 1 dem 1.1.1900** gleichgesetzt. Alle weiteren Tage werden entsprechend um 1 erhöht. Der 1. Februar 1900 ist zum Beispiel als **fortlaufende Zahl** „32“ gespeichert. Somit kann man durch einfache Subtraktionen **Datumsdifferenzen** ermitteln. In Excel für Windows wird dieses System **1900-Datumssystem<sup>1</sup>** genannt.

<sup>1</sup> In Excel für **Macintosh** ist das erste Jahr **1904**. Wenn Sie in Excel für Windows ein in Excel für Macintosh erstelltes Dokument öffnen, wird automatisch das 1904-Datumssystem verwendet. So ändern Sie das Datumssystem: REGISTER „DATEI“ → „OPTIONEN“ → „ERWEITERT“ → KONTROLLKÄSTCHEN „1904-DATUMSWERTE VERWENDEN“.


## NETTOARBEITSTAGE

Gibt die **Anzahl der Arbeitstage in einem Zeitintervall** zurück. Nicht zu den Arbeitstagen gezählt werden Wochenenden (Samstag und Sonntag) sowie die Tage, die als Ferien (Feiertage) angegeben sind.

Syntax	
<b>NETTOARBEITSTAGE(Ausgangsdatum;Enddatum;[Freie_Tage])</b>	

Argumente	Beschreibung
<b>Ausgangsdatum</b> (erforderlich)	Ein Datum, das das <b>Ausgangsdatum</b> angibt.
<b>Enddatum</b> (erforderlich)	Ein Datum, das das <b>Enddatum</b> angibt.
<b>Freie_Tage</b> (optional)	Alle Arten von <b>arbeitsfreien Tagen</b> (z. B. staatliche oder regionale Feiertage und Freischichten).

Beispiel:



	A	B	C	D	E
1	<b>Ausgangsdatum</b>			21.06.2017	
2	<b>Enddatum</b>			20.08.2017	
3	<b>Freie Tage (Samstag und Sonntag freie Tage)</b>			15.08.2017	
4	<b>Arbeitstage bis zum Enddatum:</b>			<b>42</b>	=NETTOARBEITSTAGE(D1;D2;D3)


## ARBEITSTAG

Gibt die **Datumsangabe als fortlaufenden Tag im Jahr zurück**, vor oder **nach einer bestimmten Anzahl von Arbeitstagen**. Nicht zu den Arbeitstagen gezählt werden Wochenenden (Samstag und Sonntag) sowie die Tage, die als Ferien (Feiertage) angegeben sind.

Syntax	
<b>ARBEITSTAG(Ausgangsdatum;Tage;[Freie_Tage])</b>	

Argumente	Beschreibung
<b>Ausgangsdatum</b> (erforderlich)	Ein Datum, das das <b>Ausgangsdatum</b> angibt.
<b>Tage</b> (erforderlich)	Die Anzahl der nicht auf ein Wochenende oder auf einen Feiertag fallenden Tage vor oder nach dem "Ausgangsdatum". <i>Ein positiver Wert für "Tage" bedeutet ein zukünftiges Datum, und ein negativer Wert ergibt ein zurückliegendes Datum.</i>
<b>Freie_Tage</b> (optional)	Alle Arten von <b>arbeitsfreien Tagen</b> (z. B. staatliche oder regionale Feiertage und Freischichten).

Beispiel:



	A	B	C	D	E
1	<b>Ausgangsdatum</b>			21.06.2017	
2	<b>Tage</b>			60	
3	<b>Freie Tage (Samstag und Sonntag freie Tage)</b>			15.08.2017	
4	<b>Datum nach Hinzufügen von 60 Arbeitstagen</b>			<b>14.09.2017</b>	=ARBEITSTAG(D1;D2;D3)

## ENDDATUM

Mit Hilfe dieser Funktion ist es möglich, **ganze Monate zu einem Datum zu addieren bzw. abzuziehen**.

### Syntax

**EDATUM(Ausgangsdatum;Monate)**

Argumente	Beschreibung
<b>Ausgangsdatum</b> (erforderlich)	Ein Datum, das das <b>Ausgangsdatum</b> angibt.
<b>Monate</b> (erforderlich)	Gibt an, <b>wie viele Monate vor oder nach dem Ausgangsdatum liegen</b> sollen. <i>Ein positiver Wert für Monate ergibt ein in der Zukunft, ein negativer Wert ein in der Vergangenheit liegendes Datum.</i>

Beispiele:

	A	B	C	D	E
1	Anfangsdatum			22.06.2017	
2	Monate			2	
3	Enddatum			<b>22.08.2017</b>	=EDATUM(D1;D2)
4					
5	Anfangsdatum			22.06.2017	
6	Enddatum			<b>22.04.2017</b>	=EDATUM(D1;-2)

